

# Qucs

---

## A Report Verilog-A Macromodel for Resistive Potentiometers

Mike Brinson

---

Copyright © 2008 Mike Brinson <mbrin72043@yahoo.co.uk>

Permission is granted to copy, distribute and/or modify this document under the terms of the GNU Free Documentation License, Version 1.1 or any later version published by the Free Software Foundation. A copy of the license is included in the section entitled "GNU Free Documentation License".

## Introduction

The resistive potentiometer is a common component in electronic systems. Unfortunately, it is often very poorly modelled by circuit simulators. A common approach is to simply treat the device as two series resistances with values set by linear or logarithmic algebraic equations. This is fine as a rough and ready approach to modelling the basic potentiometer function. However, it fails to acknowledge the fact that potentiometers are much more complex devices that involve not only electrical characteristics but also mechanical rotational features. Furthermore, potentiometers are subject to a number of errors such as taper conformity and linearity. This report presents a more realistic model of a resistive potentiometer and shows how Qucs can be used to develop practical models of this fundamental component that give an order of magnitude improvement over the commonly used two resistor models. Finally, the notes show how a Qucs subcircuit of the potentiometer can be coded as a compact Verilog-A macromodel.

## The two resistor potentiometer model

The schematic for a two resistor model of a linear resistive potentiometer is shown in Fig. 1. Sweep parameter *Position* is used to control the movement of the potentiometer central output terminal. A range of zero to one covers the movement of the output terminal from one end of the potentiometer to the other. This model acts more like a sliding arm potentiometer rather than the more common rotary type. Notice the use of non-integer values 1.00001 and 0.00001 to ensure that R1 or R2 do not become zero in value when the mid-terminal is slid to either end of the potentiometer. Changing equations R1eqn and R2eqn allows different laws to be specified that relate the values of resistors R1 and R2 to the position of the mid-terminal.




Figure 1: Basic two resistor model of a linear resistive potentiometer

## Potentiometer types

The basic types of resistive potentiometer are the rotary and slider forms<sup>1</sup>. These are constructed from wire, cement or conductive plastic resistive strips attached to a suitable non-conducting holder with fixed contacts at the ends and a moveable wiper contact. The moveable contact either traverses in a rotary direction or in a straight line. The resistive strip has either a constant width or is shaped depending on the type of potentiometer, giving either constant resistance or changing resistance per unit length along the length of a potentiometer. Probably the most common of the non-linear types of potentiometer is the logarithmic potentiometer. Figure 2 illustrates how linear, logarithmic and inverse logarithmic device laws vary as a function of wiper contact angle. Logarithmic potentiometers are sometimes called audio potentiometers due to their use as volume control devices in audio amplifiers. Potentiometers with logarithmic resistive laws are necessary because the human ear responds to the logarithm of loudness, making control of amplifier output level smoother with a logarithmic potentiometer rather than the simpler linear device. The shape of the potentiometer response law is often called the potentiometer taper. Today linear potentiometers are the most common type made by different manufacturers. One reason for this is the cost of manufacturing a true logarithmic law device. In many cases so called logarithmic potentiometers are in fact simpler devices constructed from two different sections of resistive material where the low angle part of the potentiometer track has a smaller resistivity than the high angle section. Logarithmic potentiometers are normally specified as 20% or 10% types, being defined as the percentage of total potentiometer resistance

<sup>1</sup>R.G. Keen, The secret life of pots, 1999, [http://www.geofex.com/Article\\_Folders/potsecrets/potscret.htm](http://www.geofex.com/Article_Folders/potsecrets/potscret.htm)

between the bottom fixed contact and the wiper contact when the wiper is moved to the middle of it's full range. Inverse logarithmic potentiometers have a similar characteristic to the logarithmic potentiometer except that potentiometer taper is reversed. This has the effect of causing large changes in output at small wiper angles and the reverse at high angles. Inverse logarithmic potentiometers have in recent years become unusual items and are no longer widely manufactured.




Figure 2: Resistive potentiometer characteristics:  $V_{in} = 1V$  DC

## Modelling potentiometer taper laws

Due to their lower cost, and availability, linear potentiometers are often chosen as a starting point when designing a circuit that requires a potentiometer with a specific taper law. The same applies to modelling potentiometers with a known resistive taper. By adding a fixed resistance, external to a potentiometer, a linear potentiometer can be converted to one with a designer specified taper. Consider the three resistor diagram shown in Fig. 3.




Figure 3: Basic two resistor potentiometer with taper resistor  $R_T$  connect between the bottom fixed contact ( $B$ ) and the wiper arm contact ( $M$ )

The transfer function for the schematic illustrated in Fig. 3 is given in equation (1).

$$\frac{V_{out}}{V_{in}} = \frac{(R_2 \parallel R_T)}{R_1 + (R_2 \parallel R_T)} = \frac{1}{1 + \frac{R_1}{R_T} + \frac{R_1}{R_T}} \quad (1)$$

Writing  $R_{pot} = R_1 + R_2$  and setting  $a = R_2/R_{pot}$  and  $TaperCoeff = R_T/R_{pot}$  yields equation (2).

$$\frac{V_{out}}{V_{in}} = \frac{1}{1 + \frac{1-a}{a} + \frac{1-a}{TaperCoeff}} = \frac{a \cdot TaperCoeff}{a - a^2 + TaperCoeff} \quad (2)$$

Where  $TaperCoeff$  is the potentiometer taper coefficient (a fixed value greater than zero) and parameter  $a$  is a measure of the wiper arm position (being in the range 0 to 1). As the taper coefficient approaches a large value (much greater than one) the potentiometer law approaches that of a simple linear potentiometer, effectively this implies that  $R_T \gg R_{pot}$ . For rotary potentiometers, parameter  $a$  can also be expressed as  $a = Angle/MaxAngle$ . In the case of a single turn potentiometer  $0 \leq Angle \leq 360$  degrees and  $MaxAngle = 360$  degrees. Figure 4 illustrates the effect of different  $TaperCoeff$  values on the potentiometer response curves. Logarithmic potentiometer response is approximated when the  $TaperCoeff$  is in the range 0.2 to 0.25.




Figure 4: Potentiometer response curves for different  $TaperCoeff$  values: curves blue to black;  $TaperCoeff = 0.1, 0.2, 0.75, 1, 2, 3, 4, 5$

## Modelling inverse logarithmic potentiometers

By placing the fixed taper resistance in parallel with resistor R1 the potentiometer response changes to the inverse logarithmic form shown in Fig. 2. Consider the three resister model shown in Fig. 5.




Figure 5: Basic two resister potentiometer with taper resister RT connect between the top fixed contact (T) and the wiper arm contact (M)

The transfer function for the schematic illustrated in Fig. 5 is given in equation (3).

$$\frac{V_{out}}{V_{in}} = \frac{R_2}{(R_1 \parallel R_T) + R_2} = \frac{R_2}{\left(\frac{R_1 \cdot R_T}{R_1 + R_T}\right) + R_2} = \frac{R_1 \cdot R_2 + R_2 \cdot R_T}{R_1 \cdot R_2 + R_1 \cdot R_T + R_2 \cdot R_T} \quad (3)$$

Writing  $R_{pot} = R_1 + R_2$  and setting  $a = R_2/R_{pot}$  and  $TaperCoeff = R_T/R_{pot}$  yields equation (4).

$$\frac{V_{out}}{V_{in}} = \frac{a - a^2 + a \cdot TaperCoeff}{a - a^2 + TaperCoeff} \quad (4)$$

Figure 6 illustrates the effect of different  $TaperCoeff$  values on the inverse potentiometer response curves. Inverse logarithmic potentiometer response is approximated when the  $TaperCoeff$  is in the range 0.2 to 0.25.




Figure 6: Inverse potentiometer response curves for different  $TaperCoeff$  values: curves blue to black;  $TaperCoeff = 0.1, 0.2, 0.75, 1, 2, 3, 4, 5$

## A Qucs subcircuit model of a resistive potentiometer

The schematic for a resistive potentiometer modelled as a Qucs subcircuit is shown in Fig. 7. This model is based on the background ideas and theory presented in the

previous sections of this report. The subcircuit schematic represents a practical resistive potentiometer, with electrical and positional characteristics set by a resistive taper coefficient, a conformity error parameter, a linearity parameter, wiper arm contact resistance and resistance temperature coefficient. The subcircuit models both single and multi-turn potentiometers of type linear, logarithmic or inverse logarithmic and indeed any other laws set by the value of parameter Taper\_Coeff. The following list outlines the function and units of the model parameters.

## Parameters

Name	Symbol	Description	Unit	Default*
Rpot	<i>Rpot</i>	Nominal device resistance	$\Omega$	1e4
Rotation	<i>Rotation</i>	Shaft/wiper arm rotation	degrees	120
Taper_Coeff	<i>TaperCoeff</i>	Resistive law taper coefficient		0
LEVEL	<i>LEVEL</i>	Device type selector		1 <sup>+</sup>
Temp	<i>Temp</i>	Circuit temperature	Celsius	26.85
Max_Rotation	<i>MaxRotation</i>	Maximum shaft/wiper rotation	degrees	240 <sup>++</sup>
Conformity	<i>Conformity</i>	Conformity error	%	0.2
Linearity	<i>Linearity</i>	Linearity error	%	0.2
Tnom	<i>Tnom</i>	Parameter measurement temperature	Celsius	26.85
Contact_Res	<i>ContactRes</i>	Wiper arm contact resistance	$\Omega$	1
Temp_Coeff	<i>TempCoeff</i>	Resistance temperature coefficient	PPM/Celsius	100 <sup>+++</sup>

\* The default parameters are for typical 10k cemet linear potentiometer.

+ Parameter LEVEL selects the potentiometer type: LEVEL=1; linear, LEVEL=2; logarithmic and LEVEL=3; inverse logarithmic.

++ For a single turn potentiometer  $240 \leq \text{MaxRotation} \leq 360$  degrees <sup>2</sup>. For multi-turn potentiometers add 360 degrees per complete shaft rotation.

+++ Potentiometer temperature coefficients are normally quoted in parts per million (PPM) per degree Celsius by manufacturers. Typical values are: cemet  $\pm 100$  PPM/Celsius, wire  $\pm 50$  PPM/Celsius and conductive plastic  $\pm 1000$  PPM/Celsius.

Potentiometer functional errors are introduced in the model illustrated in Fig. 7 by adding conformity and linearity factors to the basic taper coefficient via equation (5).

---

<sup>2</sup>The value of parameter MaxRotation is normally set by the position of an end stop at the end of the resistive track.




$$Tpcoeff = TaperCoeff + \frac{Conformity + Linearity \cdot \sin(RadAngle)}{100} \quad (5)$$

Where  $RadAngle = Rotation \cdot \pi/180$  is the shaft rotation angle in radians. Absolute conformity error is defined as the maximum deviation of the taper function characteristic from an ideal theoretical characteristic. Conformity error is normally specified in percentage and can have plus or minus values. Absolute linearity has a similar definition. However, it often shows sinusoidal characteristics<sup>3</sup> due to shaft rotation, particularly with multi-turn potentiometers. Contact resistance has been introduced into the model to take account of any resistance that exists from the wiper terminal to the contact on the potentiometer resistive track. With sprung loaded contacts this resistance should be small and can often be neglected. Effects of temperature on the model performance are included via a simple linear temperature coefficient which controls the change in device resistance as temperature varies. A potentiometer test circuit and the response of a multi-turn linear potentiometer with zero conformity and linearity errors are given in Fig. 8. Multi-turn potentiometers are modelled by setting the shaft rotation parameter to multiples of 360 degrees plus slightly less angle for the final rotation. Figure 9 illustrates a test circuit for comparing the performance of log and inverse log single-turn potentiometers. The response curves show both simulation results and those obtained by fitting logarithmic curves to the simulated output data. With the Taper\_Coeff parameter set to 0.2, remarkably good agreement is observed when the simulation data is compared to the ideal logarithmic performance. The test circuit and simulation curves illustrated in Fig. 10 demonstrate the effects that conformity and linearity parameters have on a multi-turn potentiometer. Normally both of these parameters are much less than one percent, making their effects difficult to observe. In Fig. 10 they have both been set at ten percent, greatly exaggerating their effect on potentiometer performance.


---

<sup>3</sup>See section 3.2: Novotechnik Potentiometer Definitions, [http://www.novotechnik.com/novotechnic\\_tech\\_ref.html](http://www.novotechnik.com/novotechnic_tech_ref.html)



Equation

```
Eqn1
Rtop=(1.000001-(Rotation/(Max_Rotation+1e-20)))*R_pot_Temp
Tpcoeff=Taper_Coeff+(Conformity+Linearity*sin(Rad_Angle))/100
Rad_Angle=Rotation*pi/180
R_pot_Temp=R_pot*(1+Temp_Coeff*(Temp-Tnom)/1e6)
Rbot=(0.000001+(Rotation/(Max_Rotation+1e-20)))*R_pot_Temp
RTB=(LEVEL==2 ? (Taper_Coeff != 0.0) ? R_pot_Temp*Tpcoeff : 1e15 : 1e15)
RTT=(LEVEL==3 ? (Taper_Coeff != 0.0) ? R_pot_Temp*Tpcoeff : 1e15 : 1e15)
```



```
POT1
R_pot=10k
Rotation=120
Taper_Coeff=0
LEVEL=1
Temp=26.85
Max_Rotation=240
Conformity=0.2
Linearity=0.2
Tnom=26.85
Temp_Coeff=100
Contact_Res=1
```

Figure 7: Qucs subcircuit of a resistive potentiometer: schematic, equations and symbol




Figure 8: Multi-turn linear potentiometer test circuit and performance curve




Figure 9: Single-turn log and inverse log potentiometer test circuit and performance curves: (1) simulated data solid curves, (2) fitted data dotted curves




Figure 10: Multi-turn log potentiometer test circuit and response curves, illustrating conformity and linearity errors: both conformity and linearity parameters are set at ten percent to clearly indicate their effects; blue curve Pot1, red curve Pot2 and black curve Pot3

## A Verilog-A resistive potentiometer model

One of the advantages of using Qucs subcircuits, and indeed EDD models, is that they allow fast interactive prototyping of new models. Once a new model is tested and functioning satisfactorily it can often be easily translated into Verilog-A code for compiling into C++ code, using the ADMS compiler, and finally compiled and linked with the main Qucs software. When prototyping new models it is advisable to use, whenever possible, circuit elements/structures that can be easily transposed to Verilog-A statements. The Verilog-A code listed in the next section is based directly on the Qucs subcircuit model of the potentiometer. In general there is a one-to-one correspondence between the different sections of the subcircuit model and the Verilog-A code. However, for completeness the Verilog-A code also includes white noise statements to account for resistor noise<sup>4</sup>. Overall the subcircuit model and the Verilog-A code give the same performance except that the Verilog-A code appears to simulate much faster. This is not surprising considering it comprises compiled and linked machine code rather than being a netlist which is interpreted during circuit simulation.

### Verilog-A model code

```
// Qucs resistive potentiometer model:
// This model can be used to construct working linear, logarithmic and inverse logarithmic
// potentiometers, or other devices with designer specified resistive taper functions.
// All required parameters can be extracted directly from
// manufacturers data sheets.
//
// The structure and theoretical background to the potentiometer
// Verilog-a model are presented in the Qucs potentiometer report.
//
// This is free software; you can redistribute it and/or modify
// it under the terms of the GNU General Public License as published by
// the Free Software Foundation; either version 2, or (at your option)
// any later version.
//
// Copyright (C), Mike Brinson, mbrin72043@yahoo.co.uk, February 2008.
//
#include "disciplines.vams"
#include "constants.vams"
//
module potentiometer (B, M, T);
  inout B, M, T;
//
  electrical B, M, T;
//
// Internal nodes
//
  electrical n1;
```

---

<sup>4</sup>The resistors comprising the potentiometer subcircuit have thermal noise built-in as a standard component feature.

```

//
'define attr(txt) (*txt*)
//
parameter real R_pot = 1e4 from [1e-6 : inf]
  'attr(info="nominal_device_resistance" unit = "Ohm");
parameter real Rotation = 120 from [0 : inf]
  'attr(info="shaft/wiper_arm_rotation" unit = "degrees");
parameter real Taper_Coeff = 0 from [0 : inf]
  'attr(info="resistive_law_taper_coefficient" );
parameter integer LEVEL = 1 from [1 : 3]
  'attr(info="device_type_selector" );
parameter real Max_Rotation = 240.0 from [0 : inf]
  'attr(info="maximum_shaft/wiper_rotation" unit = "degrees");
parameter real Conformity = 0.2 from [-inf : inf]
  'attr(info="conformity_error" unit = "%");
parameter real Linearity = 0.2 from [-inf : inf]
  'attr(info="linearity_error" unit = "%");
parameter real Contact_Res = 1 from [1e-6 : inf]
  'attr(info="wiper_arm_contact_resistance" unit = "Ohm");
parameter real Temp_Coeff = 100 from [0 : inf]
  'attr(info="resistance_temperature_coefficient" unit = "PPM/Celsius");
parameter real Thom = 26.85 from [-273 : inf]
  'attr(info="parameter_measurement_temperature" unit = "Celsius");
//
real Rad_Angle, R_pot_Temp, Rtop, Rbot, Tpcoeff, Rcontact;
real RTB, RTT, error_term;
real fourkt;
//
analog begin
//
// Model equations
//
Rcontact=Contact_Res+1e-6;
Rad_Angle=Rotation*'M_PI/180;
R_pot_Temp=(R_pot+1e-6)*(1+Temp_Coeff*($temperature-Thom)/1e6);
Tpcoeff=Taper_Coeff+(Conformity+Linearity*sin(Rad_Angle))/100;
error_term=(1+(Conformity+Linearity*sin(Rad_Angle))/100);
//
case (LEVEL)
  2: begin
    RTB=R_pot_Temp*Tpcoeff;
    RTT=1e15;
    Rtop=(1.000001-(Rotation/(Max_Rotation+1e-20))*R_pot_Temp;
    Rbot=(0.000001+(Rotation/(Max_Rotation+1e-20))*R_pot_Temp;
  end
  3: begin
    RTB=1e15;
    RTT=R_pot_Temp*Tpcoeff;
    Rtop=(1.000001-(Rotation/(Max_Rotation+1e-20))*R_pot_Temp;
    Rbot=(0.000001+(Rotation/(Max_Rotation+1e-20))*R_pot_Temp;
  end
  default: begin
    RTB=1e15;
    RTT=1e15;
    Rtop=(1.000001-(Rotation/(Max_Rotation+1e-20))*R_pot_Temp*error_term;
    Rbot=(0.000001+(Rotation/(Max_Rotation+1e-20))*R_pot_Temp*error_term;
  end
endcase
//
if (Taper_Coeff == 0.0)

```

```

begin
    RTB=1e15;
    RTT=1e15;
    Rtop=(1.000001-(Rotation/(Max_Rotation+1e-20)))*R_pot_Temp*error_term;
    Rbot=(0.000001+(Rotation/(Max_Rotation+1e-20)))*R_pot_Temp*error_term;
end
//
// Macromodel
//
I(T, n1) <+ V(T, n1)/Rtop;
I(T, n1) <+ V(T, n1)/RTT;
I(B, n1) <+ V(B, n1)/Rbot;
I(B, n1) <+ V(B, n1)/RTB;
I(M, n1) <+ V(M, n1)/Rcontact;
//
// Noise contributions
//
fourkt=4.0*‘P_K*$temperature;
I(T, n1) <+ white_noise(fourkt/Rtop, "thermal");
I(T, n1) <+ white_noise(fourkt/RTT, "thermal");
I(B, n1) <+ white_noise(fourkt/Rbot, "thermal");
I(B, n1) <+ white_noise(fourkt/RTB, "thermal");
I(M, n1) <+ white_noise(fourkt/Rcontact, "thermal");

//
end
endmodule

```

The ADMS syntax is a subset of Verilog-A. Allowed language structures are outlined in a SYNTAX-SUPPORTED file which can be downloaded from <http://mot-adms.sourceforge.net>. Recent news concerning the ADMS Verilog-A compiler can also be found on the ADMS Website Wiki.

## End note

Resistive potentiometers are fundamental components in electronic system design. They deserve due attention when forming part of a circuit simulation package. This report introduces a number of basic properties of potentiometric devices and outlines how they can be accurately and efficiently simulated by Qucs. While writing this report I have tried to demonstrate how practical resistive potentiometer models can be constructed from a set of basic concepts. Qucs subcircuits and component defining equations form the fundamental tools for modelling a potentiometer. Both encourage interactive model development and the subsequent conversion of their properties into Verilog-A code. My thanks to Stefan Jahn for all his help and encouragement during the period I have been working on the potentiometer model and writing this report.